Proyecto

```
using Graphs, Random, Plots, Statistics, StatsPlots, GraphMakie, ColorSchemes, Makie
, CairoMakie, FileIO, ImageMagick, Images, PlutoUI, Distributions
```

Modelo 1

Competencia de dos ideas opuestas

(Parámetros globales)

Consideraciones del modelo:

1. Estructura de la red social: Representada como un grafo donde los nodos son agentes y las aristas son conexiones sociales.

2. Estados de los agentes: Cada agente puede estar en uno de tres estados: creer en una idea ϕ, creer en la idea contraria $\neg\phi$, o ser indiferente $\perp$.

3. Transiciones entre estados: Los agentes pueden cambiar de estado basado en los mensajes que reciben de sus vecinos.

- Hay dos probabilidades/parámetros clave: λ (probabilidad de adoptar una idea) y μ (probabilidad de abandonar una idea).

4. Proceso de difusión:

- En cada paso, un agente es seleccionado al azar para transmitir su creencia.
- Los agentes conectados a este agente pueden cambiar su estado basado en el mensaje recibido.

6. Simulación Monte Carlo: Realizamos múltiples ejecuciones del modelo para obtener resultados estadísticamente significativos.

Implementación:

Definimos los posibles estados de cada individuo/agente, usamos un tipo enumerado `State` para representar los posibles estados de los agentes.

```
@enum State phi notphi ind
```

Definimos la estructura `Agent`, que cuenta con un número de identificación y el estado del agente. La estructura `Agent` es mutable, lo que permite cambiar su estado.

```
mutable struct Agent
    id::Int
    state::State
end
```

Función que simula el recibimiento de un mensaje y que, según los parámetros, el agente decidirá si adoptar o no la creencia de quien envió el mensaje (en caso de que el agente sea indiferente). En caso de que cuente con un estado distinto al mensaje, este decidirá si seguir con su creencia actual o pasar a un estado de indiferencia.

```
receive_message! (generic function with 1 method)
```

```
function receive_message!(agent::Agent, message::State, λ::Float64, μ::Float64)
    if agent.state == ind
        if rand() < λ
            agent.state = message
        end
    elseif agent.state != message
        if rand() < μ
            agent.state = ind
        end
    end
end
```

Definimos la estructura `SocialNetwork`, compuesta por el grafo de la estructura y un vector con los agentes de la misma:

```
struct SocialNetwork
    graph::SimpleGraph
    agents::Vector{Agent}
end
```

Función para inicializar una red social, con todos los agentes en estado indiferente:

```
socialNetwork (generic function with 1 method)
```

```
function socialNetwork(num_agents::Int)
    graph = barabasi_albert(num_agents, 3) # cambiar esto según  el tipo de grafo
    agents = [Agent(i, ind) for i in 1:num_agents]
    SocialNetwork(graph, agents)
end
```

Para este caso en particular, generaremos la red social utilizando el algoritmo Barabási-Albert. Para ver más tipos de grafos visitar esto.

Implementamos una función para setear los estados semilla:

```
set_seed_nodes! (generic function with 1 method)
```

```
function set_seed_nodes!(network::SocialNetwork, phi_seed::Int, notphi_seed::Int)
    network.agents[phi_seed].state = phi
    network.agents[notphi_seed].state = notphi
end
```

En este modelo consideramos dos nodos semillas: uno con el estado ϕ y uno con el estado $\neg\phi$.

Implementamos una función que simula un paso de la cadena, en este se elige un nodo aleatorio y, en caso de que dicho nodo no sea indiferente, se comparte su creencia con sus vecinos:

```
simulate_step! (generic function with 1 method)
```

```
function simulate_step!(network::SocialNetwork, λ::Float64, μ::Float64)
    sender = rand(network.agents)
    if sender.state != ind # verificamos que quien envía no sea indiferente
        for neighbor in neighbors(network.graph, sender.id)
            receive_message!(network.agents[neighbor], sender.state, λ, μ)
        end
    end
end
```

Algo que intentaremos más adelante es mejorar la manera en la que elegimos el nodo aleatorio que manda el mensaje, pues tiene más sentido que se elijan estados no indiferentes.

La siguiente función nos permite contar el número de agentes adeptos a cada uno de los estados:

```
count_beliefs (generic function with 1 method)
```

```
function count_beliefs(network::SocialNetwork)
    beliefs = Dict(s => count(a -> a.state == s, network.agents) for s in
instances(State))
    return beliefs
end
```

Finalmente, implementamos una función que nos permite correr la simulación completa:

```
run_simulation (generic function with 1 method)
```

```
function run_simulation(num_agents::Int, num_steps::Int, λ::Float64, μ::Float64, num_runs::Int)
    results = []
    for _ in 1:num_runs
        network = socialNetwork(num_agents)
        set_seed_nodes!(network, 1, 2)  # Se toma 1 y 2 como nodos semillas

        run_results = []
        for step in 1:num_steps
            simulate_step!(network, λ, μ)
            push!(run_results, count_beliefs(network))
        end

        push!(results, run_results)
    end
    return results
end
```

Veamos un ejemplo de uso:

```
sim_results =
```

```
[[Dict(phi::State = 0 ⇒ 1, ind::State = 2 ⇒ 98, notphi::State = 1 ⇒ 1), Dict(phi::Stat
```

```
sim_results = run_simulation(100, 1000, 0.5, 0.5, 1000)
```

Implementamos una función para visualizar los resultados. Esta función crea un gráfico de líneas que muestra cómo evoluciona el número esperado de agentes en cada estado (ϕ, $\neg\phi$, $\bot$) a lo largo de los pasos de la simulación.

```
plot_belief_evolution (generic function with 1 method)
```

```
function plot_belief_evolution(results)
    num_steps = length(results[1])
    num_runs = length(results)

    phi_means = [mean([run[step][phi] for run in results]) for step in 1:num_steps]
    notphi_means = [mean([run[step][notphi] for run in results]) for step in 1:num_steps]
    ind_means = [mean([run[step][ind] for run in results]) for step in 1:num_steps]

    Plots.plot(1:num_steps, [phi_means notphi_means ind_means],
         label=["φ" "¬φ" "⊥"],
         title="Evolución de creencias",
         xlabel="Pasos de simulación",
         ylabel="Número promedio de agentes",
         lw=2)
end
```

Y una función para visualizar el estado final de la cadena. Esta función crea un diagrama que muestra la distribución del número de agentes en cada estado al final de las simulaciones.

```
plot_final_distribution (generic function with 1 method)
```

```
function plot_final_distribution(results)
    final_states = [last(run) for run in results]

    phi_counts = [state[phi] for state in final_states]
    notphi_counts = [state[notphi] for state in final_states]
    ind_counts = [state[ind] for state in final_states]

    means = [mean(phi_counts), mean(notphi_counts), mean(ind_counts)]
    stds = [std(phi_counts), std(notphi_counts), std(ind_counts)]

    labels = ["φ", "¬φ", "⊥"]

    p = Plots.bar(labels, means,
        yerr=stds,
        title="Distribución final de creencias",
        ylabel="Número promedio de agentes",
        legend=false,
        color=:orange,
        alpha=0.6)
end
```






```
begin
    p1 = plot_belief_evolution(sim_results)
    p2 = plot_final_distribution(sim_results)
    Plots.plot(p1, p2, layout=(2,1), size=(950,1000))
end
```

La anterior gráfica muestra la evolución del número esperado de agentes con la creencia ϕ, $\neg\phi$ e $\bot$. Del total de pasos que se realizaron, no en todos se envió un mensaje desde un nodo a otro, pues si en un paso se elegía un agente indiferente, este no hacía nada. Veamos en promedio cuantos mensajes se enviaban después del total de simulaciones, esto para darnos una idea del número de mensajes que se requieren para que la cadena converja (empiricamente a un estado en el que la mitad de los

agentes cree ϕ y la otra mitad cree $\neg\phi$). Para esto implementamos la función `count_mean_messages_sent`.

```
count_mean_messages_sent (generic function with 1 method)
```

```
function count_mean_messages_sent(simulation_results)
    num_runs = length(simulation_results)
    messages_per_run = zeros(Int, num_runs)

    for (run, result) in enumerate(simulation_results)
        messages = 0
        prev_state = result[1]
        for state in result[2:end]
            if state != prev_state
                messages += 1
            end
            prev_state = state
        end
        messages_per_run[run] = messages
    end

    mean_messages = mean(messages_per_run)
    return mean_messages, messages_per_run
end
```

```
begin
mean_messages, messages_per_run = count_mean_messages_sent(sim_results)
    println("Media de mensajes enviados: $(round(mean_messages, digits=2))")
    println("Rango de mensajes enviados: $(minimum(messages_per_run)) -
$(maximum(messages_per_run))")
end
```

```
Media de mensajes enviados: 74.86
Rango de mensajes enviados: 39 - 475
```

Finalmente, implementaremos una función para visualizar una de las simulaciones:

```
visualize_network_evolution (generic function with 1 method)
```

```
function visualize_network_evolution(simulation_results, network)
    layout = GraphMakie.spring(network.graph)

    fig = Figure(size = (800, 600))
    ax = GraphMakie.Axis(fig[1, 1])

    function update_colors(i)
        return map(simulation_results[i]) do state
            if state == phi
                :blue
            elseif state == notphi
                :red
            else  # state == ind
                :green
            end
        end
    end

    node_colors = Observable(update_colors(1))
    p = graphplot!(ax, network.graph,
        layout = layout,
        node_color = node_colors,
        node_size = 15,
        edge_width = 1,
        edge_color = :gray80
    )

    legend_elements = [
        MarkerElement(color = :blue, marker = :circle, markersize = 15),
        MarkerElement(color = :red, marker = :circle, markersize = 15),
        MarkerElement(color = :green, marker = :circle, markersize = 15)
    ]
    legend_labels = ["φ", "¬φ", "⊥"]
    Legend(fig[1, 2], legend_elements, legend_labels, "Creencias")

    ax.title = "Evolución de la Red Social y Distribución de Creencias"

    slider = PlutoUI.Slider(1:length(simulation_results), default=1, show_value=true)

    function update_viz(i)
        node_colors[] = update_colors(i)
    end

    return fig, slider, update_viz
end
```

```
count_messages (generic function with 1 method)
```

```
# para contar los mensajes de una única simulación:
function count_messages(simulation_results)
    messages = 0
    prev_state = simulation_results[1]
    for state in simulation_results[2:end]
        if state != prev_state
            messages += 1
        end
        prev_state = state
    end
    return messages
end
```

Corremos una única simulación y la visualizamos:

```
run_and_visualize_simulation (generic function with 1 method)
```

```
function run_and_visualize_simulation(num_agents::Int, num_steps::Int, λ::Float64, μ::Float64)
    network = socialNetwork(num_agents)
    set_seed_nodes!(network, 1, 2)

    simulation_results = []
    for _ in 1:num_steps
        push!(simulation_results, [agent.state for agent in network.agents])
        simulate_step!(network, λ, μ)
    end

    messages_sent = count_messages(simulation_results)
    println("Número de mensajes enviados: $messages_sent")

    fig, slider, update_viz = visualize_network_evolution(simulation_results, network)

    return fig, slider, update_viz
end
```

Veamos una simulación: (distinta de las obtenidas en el resultado anterior)

```
fig, slider, update_viz = run_and_visualize_simulation(100, 1000, 0.5, 0.5);
```

```
Número de mensajes enviados: 255
```



Modelo 2

Competencia de dos ideas opuestas

(Parámetros locales)

Consideraciones del modelo:

Vamos a modificar el **modelo 1** para que cada agente tenga sus propios parámetros λ y μ. Esto podría permitir una mayor diversidad en el comportamiento de los agentes y potencialmente resultados más interesantes en la simulación.

Implementación:

Los estados posibles de cada agente se conservan: ϕ, $\neg\phi$ y $\bot$.

Modificamos la estructura `Agent` añadiendo los parámetros individuales de cada agente:

```
mutable struct Agent2
    id::Int
    state::State
    λ::Float64  # Probabilidad individual de adoptar una idea
    μ::Float64  # Probabilidad individual de abandonar una idea
end
```

La estructura de la red social se conserva, incluyendo su grafo y un vector con los agentes de la red.

```
struct SocialNetwork2
    graph::SimpleGraph
    agents::Vector{Agent2}
end
```

La función que simula el recibimiento de un mensaje ahora utiliza los parámetros λ y μ específicos de cada agente.

```
receive_message2! (generic function with 1 method)
```

```
function receive_message2!(agent::Agent2, message::State)
    if agent.state == ind
        if rand() < agent.λ
            agent.state = message
        end
    elseif agent.state != message
        if rand() < agent.μ
            agent.state = ind
        end
    end
end
```

Actualizamos la función `socialNetwork`: Ahora toma rangos para λ y μ, y asigna valores aleatorios dentro de esos rangos a cada agente.

```
socialNetwork2 (generic function with 1 method)
```

```
function socialNetwork2(num_agents::Int, λ_range::Tuple{Float64,Float64},
μ_range::Tuple{Float64,Float64})
    graph = barabasi_albert(num_agents, 3)
    agents = [Agent2(i, ind, rand(Uniform(λ_range...)), rand(Uniform(μ_range...)))
for i in 1:num_agents]
    SocialNetwork2(graph, agents)
end
```

Función para establecer los nodos semilla:

```
set_seed_nodes2! (generic function with 1 method)
```

```
function set_seed_nodes2!(network::SocialNetwork2, phi_seed::Int, notphi_seed::Int)
    network.agents[phi_seed].state = phi
    network.agents[notphi_seed].state = notphi
end
```

Para simular un paso de la cadena:

```
simulate_step2! (generic function with 1 method)
```

```
function simulate_step2!(network::SocialNetwork2)
    sender = rand(network.agents)
    if sender.state != ind
        for neighbor in neighbors(network.graph, sender.id)
            receive_message2!(network.agents[neighbor], sender.state)
        end
    end
end
```

Para contar el número de agentes adeptos a cada uno de los estados:

```
count_beliefs2 (generic function with 1 method)
```

```
function count_beliefs2(network::SocialNetwork2)
    beliefs = Dict(s => count(a -> a.state == s, network.agents) for s in
instances(State))
    return beliefs
end
```

Actualizamos las funciones de simulación: `run_simulation` y `run_and_visualize_simulation` ahora toman rangos para λ y μ en lugar de valores fijos.

```
run_simulation2 (generic function with 1 method)
```

```
function run_simulation2(num_agents::Int, num_steps::Int,
λ_range::Tuple{Float64,Float64}, μ_range::Tuple{Float64,Float64}, num_runs::Int)
    results = []
    for _ in 1:num_runs
        network = socialNetwork2(num_agents, λ_range, μ_range)
        set_seed_nodes2!(network, 1, 2)
        run_results = []
        for step in 1:num_steps
            simulate_step2!(network)
            push!(run_results, count_beliefs2(network))
        end
        push!(results, run_results)
    end
    return results
end
```

```
run_and_visualize_simulation2 (generic function with 1 method)
```

```
function run_and_visualize_simulation2(num_agents::Int, num_steps::Int,
λ_range::Tuple{Float64,Float64}, μ_range::Tuple{Float64,Float64})
    network = socialNetwork2(num_agents, λ_range, μ_range)
    set_seed_nodes2!(network, 1, 2)
    simulation_results = []
    for _ in 1:num_steps
        push!(simulation_results, [agent.state for agent in network.agents])
        simulate_step2!(network)
    end

    messages_sent = count_messages(simulation_results)
    println("Número de mensajes enviados: $messages_sent")

    fig, slider, update_viz = visualize_network_evolution(simulation_results, network)
    return fig, slider, update_viz
end
```

```
sim_results2 =
```

```
[[Dict(phi::State = 0 ⇒ 1, ind::State = 2 ⇒ 98, notphi::State = 1 ⇒ 1), Dict(phi::Stat
```

```
sim_results2 = run_simulation2(100, 1000, (0.3, 0.7), (0.3, 0.7), 100)
```






```
begin
    p1_ = plot_belief_evolution(sim_results2)
    p2_ = plot_final_distribution(sim_results2)
    Plots.plot(p1_, p2_, layout=(2,1), size=(950,1000))
end
```

```
begin
mean_messages2, messages_per_run2 = count_mean_messages_sent(sim_results2)
    println("Media de mensajes enviados: $(round(mean_messages2, digits=2))")
    println("Rango de mensajes enviados: $(minimum(messages_per_run2)) -
$(maximum(messages_per_run2))")
end
```

```
Media de mensajes enviados: 79.52
Rango de mensajes enviados: 42 - 471
```

Veamos una simulación: (distinta de las obtenidas en el resultado anterior)

```
fig2, slider2, update_viz2 = run_and_visualize_simulation2(10, 100, (0.3, 0.7),
(0.3, 0.7));
```

```
Número de mensajes enviados: 10
```

1



Modelo 3

Difusión de un mensaje

Consideraciones del modelo:

Este tercer modelo que se enfoca en la velocidad de propagación de un mensaje, considerando solo dos estados: informed (mensaje recibido) e uninformed (mensaje no recibido). Este modelo simplificado nos permitirá observar cómo se propaga un único mensaje a través de la red.

1. Solo dos estados: informed y uninformed.
2. Consideramos el parámetro `spread_probability` en la estructura Agent. Este parámetro ahora representa la probabilidad de que un agente decida compartir el mensaje con sus vecinos.
3. Implementamos la función `attempt_spread!` para reflejar este cambio. Ahora, cuando un agente informado intenta compartir el mensaje, usa su spread_probability para decidir si lo hace o no.
4. Actualizamos la función `simulate_step!` para que todos los agentes informados intenten compartir el mensaje con todos sus vecinos en cada paso.
5. Ajustamos las funciones de simulación y visualización para reflejar estos cambios.

Algunas cosas por ver:

- Cómo la distribución de las probabilidades de propagación afecta la velocidad general de propagación del mensaje.
- Si hay un umbral crítico de probabilidad de propagación por debajo del cual el mensaje no se propaga eficazmente.
- Cómo la estructura de la red interactúa con las probabilidades de propagación individuales.
- Si emergen "super-propagadores" (nodos que son particularmente efectivos en propagar el mensaje debido a su alta probabilidad de propagación y/o su posición en la red).

Implementación:

Definimos los posibles estados de cada individuo/agente, usamos un tipo enumerado `State` para representar los posibles estados de los agentes.

```
1 @enum State3 informed uninformed
```

Definimos la estructura que representa a un agente, esta cuenta con un id, el estado y la probabilidad que tiene el agente de esparcir el mensaje:

```
mutable struct Agent3
    id::Int
    state::State3
    spread_probability::Float64  # Probabilidad de decidir esparcir el mensaje
end
```

Definimos la estructura que representará la red social para este modelo:

```
struct SocialNetwork3
    graph::SimpleGraph
    agents::Vector{Agent3}
end
```

Definimos la siguiente función para simular el envío del mensaje desde un agente informado a uno no informado:

```
attempt_spread! (generic function with 1 method)
```

```
function attempt_spread!(sender::Agent3, receiver::Agent3)
    if sender.state == informed && receiver.state == uninformed
        if rand() < sender.spread_probability
            receiver.state = informed
        end
    end
end
```

La siguiente función nos permite inicializar la red social:

```
socialNetwork3 (generic function with 1 method)
```

```
function socialNetwork3(num_agents::Int,
spread_probability_range::Tuple{Float64,Float64})
    graph = barabasi_albert(num_agents, 3)
    agents = [Agent3(i, uninformed, rand(Uniform(spread_probability_range...))) for
i in 1:num_agents]
    SocialNetwork3(graph, agents)
end
```

La siguiente función sirve para establecer los nodos semilla:

```
set_seed_node3! (generic function with 1 method)
```

```
function set_seed_node3!(network::SocialNetwork3, seed::Int)
    network.agents[seed].state = informed
end
```

Para simular un paso de la cadena. En este todos los agentes informados intentan compartir el mensaje con cada uno de sus vecinos.

```
simulate_step3! (generic function with 1 method)
```

```
function simulate_step3!(network::SocialNetwork3)
    for agent in network.agents
        if agent.state == informed
            for neighbor_id in neighbors(network.graph, agent.id)
                attempt_spread!(agent, network.agents[neighbor_id])
            end
        end
    end
end
```

Para contar el número de agentes informados:

```
count_informed (generic function with 1 method)
```

```
function count_informed(network::SocialNetwork3)
    count(agent -> agent.state == informed, network.agents)
end
```

Para correr la simulación:

```
run_simulation3 (generic function with 1 method)
```

```
function run_simulation3(num_agents::Int, num_steps::Int,
spread_probability_range::Tuple{Float64,Float64}, num_runs::Int)
    results = []
    for _ in 1:num_runs
        network = socialNetwork3(num_agents, spread_probability_range)
        set_seed_node3!(network, 1)
        run_results = [count_informed(network)]
        for _ in 1:num_steps
            simulate_step3!(network)
            push!(run_results, count_informed(network))
        end
        push!(results, run_results)
    end
    return results
end
```

A continuación, implementamos las funciones para visualizar los resultados:

plot_message_spread (generic function with 1 method)

```
function plot_message_spread(results)
    num_steps = length(results[1])
    num_runs = length(results)
    mean_informed = [mean([run[step] for run in results]) for step in 1:num_steps]
    std_informed = [std([run[step] for run in results]) for step in 1:num_steps]

    Plots.plot(0:(num_steps-1), mean_informed,
        ribbon=std_informed,
        fillalpha=0.3,
        title="Propagación del Mensaje",
        xlabel="Pasos de simulación",
        ylabel="Número de agentes informados",
        label="Media ± Desv. Estándar",
        legend=:bottomright)
end
```

visualize_network_evolution_ (generic function with 1 method)

```
function visualize_network_evolution_(simulation_results, network)
    layout = GraphMakie.spring(network.graph)
    fig = Figure(size = (800, 600))
    ax = GraphMakie.Axis(fig[1, 1])

    function update_colors(i)
        return [state == informed ? :red : :blue for state in simulation_results[i]]
    end

    node_colors = Observable(update_colors(1))
    graphplot!(ax, network.graph,
        layout = layout,
        node_color = node_colors,
        node_size = 15,
        edge_width = 1,
        edge_color = :gray80
    )

    legend_elements = [
        MarkerElement(color = :red, marker = :circle, markersize = 15),
        MarkerElement(color = :blue, marker = :circle, markersize = 15)
    ]
    legend_labels = ["Informado", "No informado"]
    Legend(fig[1, 2], legend_elements, legend_labels, "Estado")

    ax.title = "Propagación del Mensaje en la Red Social"

    slider = PlutoUI.Slider(1:length(simulation_results), default=1, show_value=true)

    function update_viz(i)
        node_colors[] = update_colors(i)
    end

    return fig, slider, update_viz
end
```

```
run_and_visualize_simulation_ (generic function with 1 method)
```

```
function run_and_visualize_simulation_(num_agents::Int, num_steps::Int,
spread_probability_range::Tuple{Float64,Float64})
    network = socialNetwork3(num_agents, spread_probability_range)
    set_seed_node3!(network, 1)
    simulation_results = [[agent.state for agent in network.agents]]
    for _ in 1:num_steps
        simulate_step3!(network)
        push!(simulation_results, [agent.state for agent in network.agents])
    end

    msf = count_messages(simulation_results)
    print("Mensajes enviados: $msf")

   fig, slider, update_viz = visualize_network_evolution_(simulation_results, network)
    return fig, slider, update_viz
end
```

```
sim_results3 =
  [[1, 4, 38, 72, 90, 98, 99, 100, 100,    more ,100], [1, 20, 49, 82, 90, 95, 98, 99, 99,
```

```
sim_results3 = run_simulation3(100, 50, (0.1, 0.5), 100)
```




```
plot_message_spread(sim_results3)
```

```
count_messages_in_simulation (generic function with 1 method)
```

```
function count_messages_in_simulation(simulation_results)
    messages = 0
    prev_state = simulation_results[1]
    for state in simulation_results[2:end]
        messages += count(prev_state .!= state)
        prev_state = state
    end
    return messages
end
```

```
expected_messages_sent_ (generic function with 1 method)
```

```
function expected_messages_sent_(results)
    num_runs = length(results)
    message_counts = zeros(Int, num_runs)

    for i in 1:num_runs
        message_counts[i] = count_messages_in_simulation(results[i])
    end

    expected_messages = mean(message_counts)
    std_dev_messages = std(message_counts)

    return expected_messages, minimum(message_counts), maximum(message_counts)
end
```

```
begin
mean_messages3, min3, max3 = expected_messages_sent_(sim_results3)
    println("Media de mensajes enviados: $(round(mean_messages3, digits=2))")
    println("Rango de mensajes enviados: $(min3) - $(max3)")
end
```

```
Media de mensajes enviados: 6.24
Rango de mensajes enviados: 4 - 10
```

Veamos una simulación: (distinta de las obtenidas en el resultado anterior)

```
fig3, slider3, update_viz3 = run_and_visualize_simulation_(50, 20, (0.1, 0.5));
```

```
Mensajes enviados: 5
```


Propagación del Mensaje en la Red Social
2
1
0
−1
−2
−2
−1
0
1
2
Estado
Informado
No informado

Experimentos:

```
1 import DataFrames, CSV
```

- Impacto de los tamaños y estructuras de las redes en la difusión de ideas bajo los dos modelos: Creamos una serie de experimentos que varíen estos parámetros y comparen los resultados.

```
run_experiment (generic function with 1 method)
```

```
function run_experiment(model_number::Int, network_sizes, network_types, params, num_steps, num_runs)
    results = DataFrames.DataFrame(
        Model = Int[],
        NetworkSize = Int[],
        NetworkType = String[],
        AvgMessagesSent = Float64[],
        FinalPhiPercentage = Float64[],
        FinalNotPhiPercentage = Float64[],
        FinalIndPercentage = Float64[]
    )

    for size in network_sizes
        for type in network_types
            messages_sent = []
            phi_percentages = []
            notphi_percentages = []
            ind_percentages = []

            for _ in 1:num_runs
                network = create_network(type, size)
                if model_number == 1
                    simulation_result = run_simulation(size, num_steps, params.λ, params.μ, 1)[1]
                elseif model_number == 2
                    simulation_result = run_simulation2(size, num_steps, params.λ_range, params.μ_range, 1)[1]
                else  # model 3
                    simulation_result = run_simulation3(size, num_steps, params.spread_probability_range, 1)[1]
                end

                push!(messages_sent, count_messages(simulation_result))
                percentages = calculate_final_percentages(simulation_result[end], model_number)
                push!(phi_percentages, percentages.phi)
                push!(notphi_percentages, percentages.notphi)
                push!(ind_percentages, percentages.ind)
            end

            avg_messages = mean(messages_sent)
            avg_phi = mean(phi_percentages)
            avg_notphi = mean(notphi_percentages)
            avg_ind = mean(ind_percentages)

            DataFrames.push!(results, (
                model_number,
                size,
                string(type),
                avg_messages,
                avg_phi,
                avg_notphi,
                avg_ind
```

```
            ))
        end
    end

    return results
end
```

```
calculate_final_percentages (generic function with 1 method)
```

```
function calculate_final_percentages(final_state, model_number)
    total_agents = sum(values(final_state))
    if model_number == 1 || model_number == 2
        return (
            phi = 100 * final_state[phi] / total_agents,
            notphi = 100 * final_state[notphi] / total_agents,
            ind = 100 * final_state[ind] / total_agents
        )
    else  # model 3
        informed = count(s -> s == informed, final_state)
        return (
            phi = 100 * informed / total_agents,
            notphi = 0.0,
            ind = 100 * (total_agents - informed) / total_agents
        )
    end
end
```

```
create_network (generic function with 1 method)
```

```
function create_network(type, size)
    if type == :barabasi_albert
        return barabasi_albert(size, 3)
    elseif type == :erdos_renyi
        return erdos_renyi(size, 6/size)
    elseif type == :watts_strogatz
        return watts_strogatz(size, 6, 0.1)
    else
        error("Tipo de red no soportado")
    end
end
```

```
(spread_probability_range = (0.1, 0.5))
```

```
begin
network_sizes = [100, 200, 300, 400, 500, 600, 700, 800, 900, 1000]
network_types = [:barabasi_albert, :erdos_renyi, :watts_strogatz]
num_steps = 10000
num_runs = 100

params1 = (λ = 0.5, μ = 0.5)
params2 = (λ_range = (0.3, 0.7), μ_range = (0.3, 0.7))
params3 = (spread_probability_range = (0.1, 0.5),)
end
```

results_model1 =

	Model	NetworkSize	NetworkType	AvgMessagesSent	FinalPhiPercentage	FinalN
1	1	100	"barabasi_albert"	84.18	50.0	50.0
2	1	100	"erdos_renyi"	69.08	52.0	48.0
3	1	100	"watts_strogatz"	69.95	47.0	53.0
4	1	200	"barabasi_albert"	134.53	47.0	53.0
5	1	200	"erdos_renyi"	160.4	57.0	43.0
6	1	200	"watts_strogatz"	134.16	54.0	46.0
7	1	300	"barabasi_albert"	190.23	50.0	50.0
8	1	300	"erdos_renyi"	205.0	50.0	50.0
9	1	300	"watts_strogatz"	206.58	44.0	56.0
10	1	400	"barabasi_albert"	252.99	50.0	50.0
	more					
30	1	1000	"watts_strogatz"	684.54	43.129	56.525

```
1 results_model1 = run_experiment(1, network_sizes, network_types, params1, num_steps,
  num_runs)
```

results_model2 =

	Model	NetworkSize	NetworkType	AvgMessagesSent	FinalPhiPercentage	FinalN
1	2	100	"barabasi_albert"	67.61	51.0	49.0
2	2	100	"erdos_renyi"	72.65	50.0	50.0
3	2	100	"watts_strogatz"	91.27	55.0	45.0
4	2	200	"barabasi_albert"	157.47	61.0	39.0
5	2	200	"erdos_renyi"	137.91	52.0	48.0
6	2	200	"watts_strogatz"	143.39	41.0	59.0
7	2	300	"barabasi_albert"	216.83	46.0	54.0
8	2	300	"erdos_renyi"	219.42	45.0	55.0
9	2	300	"watts_strogatz"	211.41	56.0	44.0
10	2	400	"barabasi_albert"	301.28	52.0	48.0
	more					
30	2	1000	"watts_strogatz"	653.54	56.971	42.988

```
results_model2 = run_experiment(2, network_sizes, network_types, params2, num_steps, num_runs)
```

```
# results_model3 = run_experiment(3, network_sizes, network_types, params3, num_steps, num_runs)
```

```
all_results =
```

	Model	NetworkSize	NetworkType	AvgMessagesSent	FinalPhiPercentage	FinalN
1	1	100	"barabasi_albert"	84.18	50.0	50.0
2	1	100	"erdos_renyi"	69.08	52.0	48.0
3	1	100	"watts_strogatz"	69.95	47.0	53.0
4	1	200	"barabasi_albert"	134.53	47.0	53.0
5	1	200	"erdos_renyi"	160.4	57.0	43.0
6	1	200	"watts_strogatz"	134.16	54.0	46.0
7	1	300	"barabasi_albert"	190.23	50.0	50.0
8	1	300	"erdos_renyi"	205.0	50.0	50.0
9	1	300	"watts_strogatz"	206.58	44.0	56.0
10	1	400	"barabasi_albert"	252.99	50.0	50.0
	more					
60	2	1000	"watts_strogatz"	653.54	56.971	42.988

```
1 all_results = vcat(results_model1, results_model2)
```

```
"network_impact_results.csv"
```

```
1 CSV.write("network_impact_results.csv", all_results)
```

```
plot_results (generic function with 1 method)
```

```
function plot_results(results)
    p1 = Plots.plot(title="Mensajes enviados vs Tamaño de red")
    p2 = Plots.plot(title="Porcentaje final de creencias vs Tamaño de red")

    for model in unique(results.Model)
        for type in unique(results.NetworkType)
            data = results[(results.Model .== model) .& (results.NetworkType .== type), :]
            Plots.plot!(p1, data.NetworkSize, data.AvgMessagesSent,
                  label="Modelo $(model) - $(type)", marker=:circle)
            Plots.plot!(p2, data.NetworkSize, data.FinalPhiPercentage,
                  label="Modelo $(model) - $(type) - φ", marker=:circle)
            # Plots.plot!(p2, data.NetworkSize, data.FinalNotPhiPercentage,
                  #label="Modelo $(model) - $(type) - ¬φ", marker=:square)
            # Plots.plot!(p2, data.NetworkSize, data.FinalIndPercentage,
                  #label="Modelo $(model) - $(type) - ⊥", marker=:diamond)
        end
    end

    Plots.xlabel!(p1, "Tamaño de red")
    Plots.ylabel!(p1, "Promedio de mensajes enviados")
    Plots.xlabel!(p2, "Tamaño de red")
    Plots.ylabel!(p2, "Porcentaje final de agentes")
    Plots.ylims!(p2, (0, 100))

    Plots.plot(p1, p2, layout=(2,1), size=(800,1000))
end
```






```
1 plot_results(all_results)
```

"/Users/tomas/DocumentosU/SextoSemestre/Cadenas de Markov/Proyecto/repo/network_impact_ana

```
1 savefig("network_impact_analysis.png")
```

- investigar el impacto de la conectividad inicial de los nodos semilla.

```
run_experiment_seed_connectivity (generic function with 1 method)
```

```
function run_experiment_seed_connectivity(model_number::Int, network_sizes, network_types, params, num_steps, num_runs, seed_strategies)
    results = DataFrames.DataFrame(
        Model = Int[],
        NetworkSize = Int[],
        NetworkType = String[],
        SeedStrategy = String[],
        AvgMessagesSent = Float64[],
        FinalPhiPercentage = Float64[],
        FinalNotPhiPercentage = Float64[],
        FinalIndPercentage = Float64[]
    )

    for size in network_sizes
        for type in network_types
            for strategy in seed_strategies
                messages_sent = []
                phi_percentages = []
                notphi_percentages = []
                ind_percentages = []

                for _ in 1:num_runs
                    if model_number == 1
                        network = socialNetwork(size)
                    elseif model_number == 2
                        network = socialNetwork2(size, params.λ_range, params.μ_range)
                    end

                    seed_nodes = select_seed_nodes(network, strategy)


                    if model_number == 1
                        set_seed_nodes!(network, seed_nodes[1], seed_nodes[2])
                        simulation_result = run_simulation(size, num_steps, params.λ, params.μ, 1)[1]
                    elseif model_number == 2
                        set_seed_nodes2!(network, seed_nodes[1], seed_nodes[2])
                        simulation_result = run_simulation2(size, num_steps, params.λ_range, params.μ_range, 1)[1]
                    end

                    push!(messages_sent, count_messages(simulation_result))
                    final_state = simulation_result[end]
                    total_agents = sum(values(final_state))
                    push!(phi_percentages, 100 * final_state[phi] / total_agents)
                    push!(notphi_percentages, 100 * final_state[notphi] / total_agents)
                    push!(ind_percentages, 100 * final_state[ind] / total_agents)
                end

                DataFrames.push!(results, (
                    model_number,
                    size,
```

```
                    string(type),
                    string(strategy),
                    mean(messages_sent),
                    mean(phi_percentages),
                    mean(notphi_percentages),
                    mean(ind_percentages)
                ))
            end
        end
    end

    return results
end
```

```
select_seed_nodes (generic function with 1 method)
```

```
function select_seed_nodes(network, strategy)
    if strategy == :high_degree
        degrees = degree(network.graph)
        sorted_nodes = sortperm(degrees, rev=true)
        return sorted_nodes[1:2]  # Seleccionar los dos nodos con mayor grado
    elseif strategy == :low_degree
        degrees = degree(network.graph)
        sorted_nodes = sortperm(degrees)
        return sorted_nodes[1:2]  # Seleccionar los dos nodos con menor grado
    elseif strategy == :random
        return rand(1:length(network.agents), 2)  # Seleccionar dos nodos al azar
    end
end

```

plot_results_seed_connectivity (generic function with 1 method)

```
function plot_results_seed_connectivity(results)
    p1 = Plots.plot(title="Mensajes enviados vs Tamaño de red")
    p2 = Plots.plot(title="Porcentaje final de φ vs Tamaño de red")

    for model in unique(results.Model)
        for type in unique(results.NetworkType)
            for strategy in unique(results.SeedStrategy)
                data = results[(results.Model .== model) .& (results.NetworkType .== type) .& (results.SeedStrategy .== strategy), :]
                Plots.plot!(p1, data.NetworkSize, data.AvgMessagesSent,
                      label="Modelo $(model) - $(type) - $(strategy)", marker=:circle)
                Plots.plot!(p2, data.NetworkSize, data.FinalPhiPercentage,
                      label="Modelo $(model) - $(type) - $(strategy)", marker=:circle)
            end
        end
    end

    Plots.xlabel!(p1, "Tamaño de red")
    Plots.ylabel!(p1, "Promedio de mensajes enviados")
    Plots.xlabel!(p2, "Tamaño de red")
    Plots.ylabel!(p2, "Porcentaje final de agentes con φ")

    Plots.plot(p1, p2, layout=(2,1), size=(800,1000))
end
```

seed_strategies = [:high_degree, :low_degree, :random]

```
seed_strategies = [:high_degree, :low_degree, :random]
```

(λ_range = (0.3, 0.7), μ_range = (0.3, 0.7))

```
begin
params1_ = (λ = 0.5, μ = 0.5)
params2_ = (λ_range = (0.3, 0.7), μ_range = (0.3, 0.7))
end
```

results_model1_seed =

	Model	NetworkSize	NetworkType	SeedStrategy	AvgMessagesSent	FinalPhiPer
1	1	100	"barabasi_albert"	"high_degree"	73.02	55.0
2	1	100	"barabasi_albert"	"low_degree"	74.14	40.0
3	1	100	"barabasi_albert"	"random"	75.33	51.0
4	1	100	"erdos_renyi"	"high_degree"	75.09	47.0
5	1	100	"erdos_renyi"	"low_degree"	81.49	38.0
6	1	100	"erdos_renyi"	"random"	70.17	54.0
7	1	100	"watts_strogatz"	"high_degree"	85.75	43.0
8	1	100	"watts_strogatz"	"low_degree"	68.47	58.0
9	1	100	"watts_strogatz"	"random"	82.84	47.0
10	1	200	"barabasi_albert"	"high_degree"	135.01	41.0
more						
90	1	1000	"watts_strogatz"	"random"	663.24	51.966

```
results_model1_seed = run_experiment_seed_connectivity(1, network_sizes,
network_types, params1_, num_steps, num_runs, seed_strategies)
```

results_model2_seed =

	Model	NetworkSize	NetworkType	SeedStrategy	AvgMessagesSent	FinalPhiPer
1	2	100	"barabasi_albert"	"high_degree"	94.23	49.0
2	2	100	"barabasi_albert"	"low_degree"	75.51	56.0
3	2	100	"barabasi_albert"	"random"	72.7	56.0
4	2	100	"erdos_renyi"	"high_degree"	73.55	45.0
5	2	100	"erdos_renyi"	"low_degree"	79.22	45.0
6	2	100	"erdos_renyi"	"random"	80.66	40.0
7	2	100	"watts_strogatz"	"high_degree"	79.16	52.0
8	2	100	"watts_strogatz"	"low_degree"	75.31	45.0
9	2	100	"watts_strogatz"	"random"	92.13	41.0
10	2	200	"barabasi_albert"	"high_degree"	144.05	50.0
	more					
90	2	1000	"watts_strogatz"	"random"	608.79	58.026

```
results_model2_seed = run_experiment_seed_connectivity(2, network_sizes,
network_types, params2_, num_steps, num_runs, seed_strategies)
```

all_results_seeds =

	Model	NetworkSize	NetworkType	SeedStrategy	AvgMessagesSent	FinalPhiPe
1	1	100	"barabasi_albert"	"high_degree"	73.02	55.0
2	1	100	"barabasi_albert"	"low_degree"	74.14	40.0
3	1	100	"barabasi_albert"	"random"	75.33	51.0
4	1	100	"erdos_renyi"	"high_degree"	75.09	47.0
5	1	100	"erdos_renyi"	"low_degree"	81.49	38.0
6	1	100	"erdos_renyi"	"random"	70.17	54.0
7	1	100	"watts_strogatz"	"high_degree"	85.75	43.0
8	1	100	"watts_strogatz"	"low_degree"	68.47	58.0
9	1	100	"watts_strogatz"	"random"	82.84	47.0
10	1	200	"barabasi_albert"	"high_degree"	135.01	41.0
more						
180	2	1000	"watts_strogatz"	"random"	608.79	58.026

```
1 all_results_seeds = vcat(results_model1_seed, results_model2_seed)
```

"seed_connectivity_impact_results.csv"

```
1 CSV.write("seed_connectivity_impact_results.csv", all_results)
```






```
1 plot_results_seed_connectivity(all_results_seeds)
```

"/Users/tomas/DocumentosU/SextoSemestre/Cadenas de Markov/Proyecto/repo/seed_connectivity_

```
1 savefig("seed_connectivity_impact_analysis.png")
```

- Investigar cambio de fases sobre el grafo reticular $k \times k$.

```
create_lattice_graph (generic function with 1 method)
```

```
function create_lattice_graph(n, m)
    # g = Grid([n, m])
    g = barabasi_albert(n*m, 3)
    return Graph(g)
end
```

```
socialNetworkLattice (generic function with 1 method)
```

```
function socialNetworkLattice(n, m)
    graph = create_lattice_graph(n, m)
    agents = [Agent(i, ind) for i in 1:nv(graph)]
    SocialNetwork(graph, agents)
end
```

```
run_simulationLattice (generic function with 1 method)
```

```
function run_simulationLattice(n, m, num_steps, λ, μ, num_runs)
    results = []
    for _ in 1:num_runs
        network = socialNetworkLattice(n, m)
        set_seed_nodes!(network, 50, 59)  # Esquinas opuestas como semillas

        for _ in 1:num_steps
            simulate_step!(network, λ, μ)
        end

        final_state = count_beliefs(network)
        total_agents = sum(values(final_state))
        push!(results, (
            phi = final_state[phi] / total_agents,
            notphi = final_state[notphi] / total_agents,
            ind = final_state[ind] / total_agents
        ))
    end
    return results
end
```

```
parameter_sweep (generic function with 1 method)
```

```
function parameter_sweep(n, m, num_steps, λ_range, μ_range, num_runs)
    results = []
    for λ in λ_range
        for μ in μ_range
            sim_results = run_simulationLattice(n, m, num_steps, λ, μ, num_runs)
            avg_results = (
                λ = λ,
                μ = μ,
                phi = mean([r.phi for r in sim_results]),
                notphi = mean([r.notphi for r in sim_results]),
                ind = mean([r.ind for r in sim_results])
            )
            push!(results, avg_results)
        end
    end
    return results
end
```

```
1000
```

```
begin
    n, m = 10, 10
    num_stepsL = 1000
    λ_rangeL = 0.0:0.05:1.0
    μ_rangeL = 0.0:0.05:1.0
    num_runsL = 1000
end
```

```
sweep_results =
  [(λ = 0.0, μ = 0.0, phi = 0.01, notphi = 0.01, ind = 0.98), (λ = 0.0, μ = 0.05, phi = 0.00
```

```
sweep_results = parameter_sweep(n, m, num_stepsL, λ_rangeL, μ_rangeL, num_runsL)
```

```
plot_phase_diagram (generic function with 1 method)
```

```
function plot_phase_diagram(results)
    λ_values = unique([r.λ for r in results])
    μ_values = unique([r.μ for r in results])

    phi_data = [r.phi for r in results]
    phi_matrix = reshape(phi_data, (length(μ_values), length(λ_values)))

    Plots.heatmap(λ_values, μ_values, phi_matrix,
            xlabel="λ", ylabel="μ", title="Proporción final de agentes con creencia
  φ",
            color=:viridis)
end
```




```
plot_phase_diagram(sweep_results)
```

```
find_critical_points (generic function with 1 method)
```

```
function find_critical_points(results)
    λ_values = unique([r.λ for r in results])
    μ_values = unique([r.μ for r in results])

    critical_points = []

    for μ in μ_values
        μ_results = filter(r -> r.μ ≈ μ, results)
        sort!(μ_results, by = r -> r.λ)

        for i in 2:length(μ_results)
            if abs(μ_results[i].phi - μ_results[i-1].phi) > 0.1  # Umbral arbitrario
                push!(critical_points, (λ = μ_results[i].λ, μ = μ))
                break
            end
        end
    end

    return critical_points
end
```

```
critical_points =
  [(λ = 0.05, μ = 0.0), (λ = 0.05, μ = 0.05), (λ = 0.1, μ = 0.1), (λ = 0.1, μ = 0.15), (λ =
```

```
critical_points = find_critical_points(sweep_results)
```

```
find_critical_lambda (generic function with 1 method)
```

```
begin
function run_simulation_fixed_mu(n, m, num_steps, λ, μ, num_runs)
    results = []
    for _ in 1:num_runs
        network = socialNetworkLattice(n, m)
        set_seed_nodes!(network, 1, n*m)

        for _ in 1:num_steps
            simulate_step!(network, λ, μ)
        end

        final_state = count_beliefs(network)
        total_agents = sum(values(final_state))
        push!(results, final_state[phi] / total_agents)
    end
    return results
end

function parameter_sweep_fixed_mu(n, m, num_steps, λ_range, μ_values, num_runs)
    results = []
    for μ in μ_values
        for λ in λ_range
            sim_results = run_simulation_fixed_mu(n, m, num_steps, λ, μ, num_runs)
            push!(results, (λ = λ, μ = μ, mean_phi = mean(sim_results), var_phi =
var(sim_results)))
        end
    end
    return results
end

function find_critical_lambda(results, μ)
    μ_results = filter(r -> r.μ ≈ μ, results)
    sort!(μ_results, by = r -> r.λ)

    max_var_index = argmax([r.var_phi for r in μ_results])
    return μ_results[max_var_index].λ
end
end
```

```
μ_valuesL =   [0.1, 0.3, 0.5, 0.7, 0.9]
```

```
μ_valuesL = [0.1, 0.3, 0.5, 0.7, 0.9]
```

```
sweep_results2 =
  [(λ = 0.0, μ = 0.1, mean_phi = 0.00973, var_phi = 2.62973e-6), (λ = 0.05, μ = 0.1, mean_ph
```

```
sweep_results2 = parameter_sweep_fixed_mu(n, m, num_stepsL, λ_rangeL, μ_valuesL,
num_runsL)
```




```
begin
Plots.plot(title="Proporción media de agentes con creencia φ vs λ")
for μ in μ_valuesL
    μ_results = filter(r -> r.μ ≈ μ, sweep_results2)
    Plots.plot!(
        [r.λ for r in μ_results],
        [r.mean_phi for r in μ_results],
        label="μ = $μ"
    )
end
Plots.xlabel!("λ")
Plots.ylabel!("Proporción media de φ")
end
```

"/Users/tomas/DocumentosU/SextoSemestre/Cadenas de Markov/Proyecto/repo/mean_phi_vs_lambda

```
savefig("mean_phi_vs_lambda.png")
```




```
begin
# Gráfico de la varianza de la proporción de agentes con creencia φ
Plots.plot(title="Varianza de la proporción de agentes con creencia φ vs λ")
for μ in μ_valuesL
    μ_results = filter(r -> r.μ ≈ μ, sweep_results2)
    Plots.plot!(
        [r.λ for r in μ_results],
        [r.var_phi for r in μ_results],
        label="μ = $μ"
    )
end
Plots.xlabel!("λ")
Plots.ylabel!("Varianza de la proporción de φ")
end
```

```
"/Users/tomas/DocumentosU/SextoSemestre/Cadenas de Markov/Proyecto/repo/var_phi_vs_lambda.
```

```
savefig("var_phi_vs_lambda.png")
```